Register Windo ws and User-Space Threads on the SP AR C

�

David Kepp el

T echnical Rep ort #91-08-0 1

Department of Computer Science and Engineering

University of W ashington

Seattle, W ashington 98195

1 August 1991

Abstract

Multiple lightweight pro cesses or thr e ads have multiple stacks, and a thread context switch moves

execution from one stack to another. On the SP ARC

1

architecture, parts of a thread’s stack can b e

cached in register windows while the thread is running. The cached data must b e ushed to memory

when the thread is susp ended. Doing the ushing b oth e�ciently and correctly can b e tricky

. This

do cument discusses the implementation of a non-preemptive user-space threads pack age under SunOS

2

.

1 In tro duction

Lightweight pro cess es executing in a single address space are called thr e ads [Birrell 89]. Conceptually

, each

of the threads of control can run indep endently and concurrently

, and, since they share one address space,

they can share data. An implementation can run a large numb er of threads on a small numb er of pro cessors

by multiplexing the threads onto the pro cessors .

There are two alternatives for threads. Systems such as Mach [Y oung et al. 87] and T opaz [Thacker et al. 88]

have kernel thr e ads. Each thread is created in the op erating system kernel and thread scheduling decisions,

such as which thread to run and on which pro cessor, are made in the kernel. An adv antage of kernel threads

is that when one thread p erforms a blo cking system call (e.g., a read that blo cks for user input), the kernel

can \salv age" the pro cess or of the blo cked thread to run another thread in the same address space. A

disadv antage is that creating, scheduling, and destroying threads are all done in the kernel, which makes

threads exp ensive b ecause all thread op erations cross kernel protection b oundaries.

An alternative is user-sp ac e thr e ads (or just user thr e ads) such as Presto [Bershad et al. 88] and F ast-

Threads [Anderson et al. 89]. User-spac e threads are managed entirely within the user address space. User

threads can b e faster b ecause they don’t cross protection b oundaries and b ecause their total state can b e

minimized for each application. User-space threads su�er from the disadv antage that when a thread executes

a blo cking kernel call, the kernel do es not know to start another thread. The blo cking call reduce s the num-

b er of pro cessors running in the address space. Many systems have user threads for simplicity

, p erformance,

p ortability

, or for compatibility with op erating systems that don’t supp ort kernel threads.

Typically

, a user-space thread context switch involves saving and restoring register v alues, including the

stack p ointer. On the SP

ARC architecture, parts of the thread’s stack may b e cached in r e gister windows.

The caching makes it di�cult to implement threads b oth e�ciently and correctly

. In addition, the register

�

This work was supp orted in part by NSF CCR-8619663, CCR-894190 by Bo eing W280638 and by Sun Microsystems

1

SP ARC is a trademark of SP

ARC International.

2

SunOS is a trademark of Sun Microsystems, Inc.

windows can only b e access ed at kernel privilege, so some op erating system supp ort is required. This

do cument discusse s the implementation of user threads on the SP

ARC pro cessor.

3

Some op erating system

services are required, but in most imp ortant asp ects the threads b ehave like user-space threads, with all

creation and scheduling decisions done in the user pro cess .

This do cument is organized as follows. First we discuss general terminology (x 2) and general issues for a

user threads pack age (x3). Then we describ e the SP

ARC stack layout and register saving mechanisms (xx 4,

5, 6). Next, we discuss optimizations to eliminate redundant loads and stores (xx 7, 8). Finally

, we present

conclusions (x 9). W e also include two app endices. One describ e s some details of the kernel implementation

of register set saving and restoring (App endix A). The other discusses tradeo�s for threads pack ages that

use the register windows to allo cate one register set p er thread (App endix B).

2 T erminology

, Bac kgroun d, and Con v en tions

It is assumed that the user is already familiar with the implementation of a basic user-space threads pack age.

This do cument assumes familiarity with the basics of SP

ARC register windows and assembly language

[Sun 91]. The key concepts include:

� Cal ler-save and c al le e-save r e gisters: Two functions (the c al ler and the c al le e) will b oth use registers.

The caller and the callee follow a proto col to ensure that the callee do es not clobb er (�ll with garbage

v alues) registers that have v alid caller data. The caller assumes that a particular set of registers will

b e used by the callee, and it is up to the caller to ensure that those registers do not have v alid data

at the pro cedure call site. These are called c al ler-save registers. Another set of registers (usually all

of the remaining registers ) are assumed to b e unused by the callee. If the callee wants to use them, it

must save them. These are c al le e-save registers [Fischer & LeBlanc 88].

� In, lo c al, out and glob al r e gisters: The SP

ARC integer registers are divided in to four groups. The in

registers hold the �rst six incoming function parameters (%i0..%i5), the frame p ointer (%i6) and the

function return address (%i7). The in registers and the lo c al registers (%l0..%l7) are callee-save registers.

The out registers are used at function calls to hold the �rst six function parameters (%o0..%o5), the

stack p ointer (%o6), and the return address for function return (%o7). The out registers are caller-save

registers. The glob al registers

4

are assigned as follows: Register %g0 is always zero. Register %g1 is a

‘very temp’ that is caller-save. Registers %g2..%g4 are reserved for application co de and may not b e

mo di�ed by libraries; whether they are caller-save or callee-save is up to the application. Registers

%g5..%g7 may b e read but not written by an application (e.g., the analog of %g2..%g4 but for use by

libraries).

� Save and R estor e: On function entry

, the callee sees the same v alues in the registers as were seen by the

caller. The save instruction causes the registers to b e remapp ed as follo ws: The in and lo cal registers

are saved to memory (logically; they may in fact b e cached in the pro cessor). The out registers are

copied to the in registers. The old out and lo cal register v alues are clobb ered. The global registers are

unchanged. On function exit, the restore instruction causes the registers to b e remapp ed again: The

in registers are copied to the out registers . The in and lo cal registers have their old v alues restored.

The global registers are unchanged.

� Over ow and Under ow: The actual implementation of save and restore is that registers are not

copied. Instead, the pro cessor has a large numb er of registers and they are reference d by a sliding

window. The windowed registers are organized so that the out registers are at the top of the window,

3

This do cument will discuss implementations that use the windows entirely for one thread, then entirely for another thread,

and so on. Some threads pack ages allo cate one register set p er thread. In these systems, the register set management issues

are di�erent. See App endices A and B.

4

In compiler terminology

, global registers are those that are allo cated over a whole function. In SP

ARC terminology

, global

registers is the name for registers %g0..%g7.

cswap (old, new)

{

push <registers>

old->sp = sp

sp = new->sp

pop <registers>

return

}

Figure 1: Typical context swap co de

the lo cals in the middle, and the ins at the b ottom. A save instruction slides the window so that the

registers at the top of the window that were visible as out registers b ecome visible at the b ottom of the

window as the in registers . The restore instruction slides the window down again. When the window

slides o� the top or b ottom of the real registers, then the pro cessor traps. The trap handler saves or

restores the register v alues as needed.

� R e gister sets/r e gister windows: A window is a set of 24 registers, comp osed of eight each in, lo cal, and

out registers. A r e gister set is 16 registers. Eight of the registers are lo cal registers and the other eight

are the in registers asso ciated with those lo cals. Note, however, that the eight in registers are the out

registers for an adjacent in/lo cal register set.

� Save Ar e a Chaining: The stack p ointer (in a window’s out register) p oints to the save area for the in

and lo cal registers. If a restore instruction causes an under ow trap, the frame p ointer in the current

register window is the stack p ointer { and thus a p ointer to the save area { for the register set that

needs to b e reloaded. That reloaded register set has a frame p ointer that is the save area for the next

register set. Register sets are thus threaded down the stack.

� L e af function optimization: A leaf function is one that calls no other functions. On the SP

ARC, some

leaf functions may b e optimized so that they op erate in the caller’s registers. A leaf function do es

not execute the save and restore instructions. Thus, a leaf function can use the out registers freely

but must spill and restore ins and lo cals if it wishes to use them. The adv antages of leaf pro cedure

optimization are that (a) the save and restore instructions don’t need to b e executed and (b) there

is no p ossibility for window over ow.

Although over ow and under ow traps only happ en for some save and restore instructions, they are

fairly exp ensive when they do happ en. In addition, an over ow forces some register sets to b e spilled,

so there will b e another trap later to restore them.

This do cument uses the term inlining to mean that the co de for a pro cedure is inserted into the co de for

another pro cedure . Inlining is also called op en-co ding, inline expansion [Fischer & LeBlanc 88], pro cedure

integration, or pro cedure hoisting.

3 Generic Thread Sw aps

A thr e ad c ontext swap is when one thread (old) is stopp ed and another thread (new) is started. A thread

context swap requires saving the state of the old thread and restoring the state of the new thread. The state

of a thread is usually just its program counter, stack p ointer, and active pro cessor registers. Figure 1 shows

prototypical thread context switch co de.

F or synchronous (non-preemptive) context swaps, only some of the registers need to b e saved. The caller-

save registers are saved b efore calling the context swap routine, so it is not neces sary to save them inside sub %sp, 8, %sp ! allocate

st %g0, [%sp+64] ! lowest-address locations

st %g0, [%sp+64+4] ! higher-address locations

Figure 2: Zeroing the top eight bytes of user stack space

st %l1, [%sp] ! Save register value.

ld [%sp], %l2 ! Load it elsewhere.

cmp %l1, %l2 ! Are they the same?

bne abort ! Jump if not.

Figure 3: Data at [%sp] may b e changed by interrupts

of cswap. In most cases, oating-p oin t and global registers are caller-save registers. Callee-save registers

must b e saved by the context swap routine. If the co de of cswap is inlined into another function, it may b e

necess ary for cswap to save caller-save registers.

In general, the registers can b e saved anywhere. There are several reasons to save them on the thread

stack:

� There must b e one register save area for each susp ended thread. There is already one stack p er thread.

� Register save space is never needed when the thread is running. Saving registers to the current top of

the stack allo ws us to also use the register save space while the thread is running.

� Register save space has to b e allo cated somewhere. By allo cating it on the top of the stack, cache

lo cality of refere nce is improved.

The remainder of this do cument assumes that all of the registers except the stack p ointer are saved on

the stack, and that the stack p ointer is used by the cswap routine to �nd the other v alues saved on the stack.

(In Figure 1, the program counter is saved implicitly b ecause all threads are restarted in the middle of the

cswap routine.)

4 SP AR C Stac k La y out

This section describ e s the SP

ARC stack layout conventions. The SP

ARC stack is di�erent than the stack

on most other machines b ecause the stack p ointer must always b e v alid and b ecause the top 64 bytes may

b e overwritten unpredictably .

The SP

ARC stack grows down and must always b e kept 8-byte aligned. The top of the stack is the

most-recently-used part, and is at the lowest addresses. The top 64 bytes of the SP

ARC stack are reserved

as a place for the kernel to save 16 4-byte registers. In this do cument, the 64-byte save region is called the

kernel window save ar e a or kwsa. On interrupts, traps, etc., the in and lo cal registers may b e saved to the

kernel window save area. The rest of the call frame is av ailable to the user.

Figure 2 shows a co de fragment that allo cates and zeros the top 8 user bytes on the stack. Note that

b ecause of the kernel window save area, the top 8 user bytes are actually 64 bytes from the real top of stack.

Since the real top of the stack can b e overwritten unpredictably by interrupts and traps, any v alue saved

to the top of the stack can b e invisibly clobb ered. Figure 3 shows user-mo de co de that uses a part of the

stack reserve d for use by the kernel. The co de will sometimes jump to abort. The jump will happ en when

(a) there is an exception such as an interrupt b etween st and ld and (b) the register window over ows

| | caller’s

+-----------------+ call frame

| kernel window | |

| save -- 64bytes | v

+-----------------+ <- fp ---

+-----------------+ ---

| | ^

| user data | |

| | top-level (callee)

+-----------------+ call frame

| kernel window | |

| save -- 64bytes | v

+-----------------+ <- sp ---

Figure 4: Register set save areas p opulate the stack

during the handling of the exception and (c) the register v alue that is written to [%sp] is di�erent than the

v alue in %l1.

Since asynchronous events can cause a write to the top of the stack, the stack p ointer %sp must always

p oint to an area that can b e used for a save. F or example, if page zero is unwritable, setting %sp to page

zero is an error, even if %sp is not dereferenc ed by the user co de.

On every pro cedure call, the frame p ointer in the callee p oints to the caller’s save area (see Figure 4).

The caller’s save area contains a saved frame p ointer that in turn p oints to its caller’s save area. Thus, each

stack is \p opulated" with kernel window save areas. In general, several register sets may b e saved at once.

Thus, when one kernel window save area is read or written, several of them may b e written.

Once again: the kernel window save areas can b e clobb ered asynchronously

. The imp ortance of this will

b e discusse d shortly .

5 Sa ving Thread Registers

This section considers tradeo�s b etween the complexity and the e�ciency of the the co de that saves the

\top-level" registers (globals, ins, lo cals, and outs). Saving cached register sets is discussed in the follo wing

section.

5.1 Regular, Leaf, or Inlined Pro cedure?

Di�erent implementations of cswap will need to save di�erent registers. Several choices are discussed b elow.

If the thread context swap routine is not treated as a leaf function [Sun 91], then only the frame p ointer

(%fp) and function return address (%i7) registers need to b e saved. As discussed in x 2, globals and oating-

p oint registers do not usually need to b e saved, b ecause most compilers and systems assume that they are

clobb ered across function calls. The in registers don’t need to b e saved b ecause the caller assumes that the

callee clobb ers them. The lo cal and out registers don’t need to b e saved b ecause we aren’t using them in

the cswap routine.

The thread context swap routine can b e optimized by making it a leaf pro cedure. Leaf pro cedure s execute

in the context (register set) of the caller; that means that the stack do esn’t grow when the cswap routine

is called. Not growing the stack reduces the numb er of register window over ows and under ows, making

call/return faster. However, since cswap executes in the caller’s registers , the in and lo cal registers need

to b e saved. Ab ove, in and lo cal register save and restore were done implicitly by the pro cedure link age sub %sp, SIZE, %sp ! Allocate a save area.

st %reg1, [%sp+64+0] ! Save a register.

st %reg2, [%sp+64+4] ! Save another register.

... ! Up to SIZE bytes of registers.

Figure 5: \Generic" save co de: allo cate space and save registers

ld [%sp+64+0], %reg1 ! Restore a register.

ld [%sp+64+4], %reg2 ! Restore another register.

... ! Up to SIZE bytes of registers.

add %sp, SIZE, %sp ! Deallocate save area.

Figure 6: \Generic" restore: fetch each register and deallo cate space

mechanism; in a leaf function, the save and restore have to b e done explicitly by cswap. However, no extra

loads or stores have b een intro duced.

If the context switch co de is inlined into the caller, then the out registers need to b e saved as well, b ecause

they may b e used as scratch registers.

5.2 Sa ving and Restoring Registers Man ually

Figure 5 shows \generic" co de to save registers. That co de:

� Allo cates stack space for saving registers. The needed size must b e rounded up to the next 8-byte size

in order to keep the stack doubleword aligned.

� Saves each register into the save area.

The generic co de for restoring registers is similar (Figure 6).

If cswap is a non-leaf pro cedure, then it will contain save and restore instructions to adjust the register

windows. As an optimization, the stack adjustment to allo cate a register save area in the old thread can

b e done by the save instruction, and the stack adjust to deallo cate the save area in the new thread can b e

done automatically by the restore instruction.

save %sp, NREGS*4, %sp

If adjacent registers are b eing saved, for instance %i0 and %i1, then the store double (std) and load

double (ldd) instructions may b e useful. They will certainly b e denser and may b e faster.

6 Sa ving Cac hed Register Sets

What has b een describ ed so far is how to save the \top-level" registers. This section describ es how to save

cached register sets to their resp ec tive kernel window save areas.

The numb er of register sets that are cached in windows is di�erent for each context swap. The numb er

of cached sets dep ends on factors such as pro cedure nesting level, recent system calls, and recent pro cess

signals. However, not only do es the numb er of sets v ary , but access ing register sets explicitly requires

sup ervisor privilege. Thus, we use a trap (op erating system entry p oint) that do es nothing but ush the

register windows out to memory

. Figure 7 shows typical usage under SunOS.

On a thread context switch, two things must happ en. First, all register sets cached in register windows

must b e ushed out to their resp e ctive kernel window save areas. Second, all register sets except the current

#include <machine/trap.h>

...

ta ST_FLUSH_WINDOWS

Figure 7: Flushing register windows under SunOS

st %sp, [%i0+SP] ! Save away ‘old thread’ handle

ta ST_FLUSH_WINDOWS ! Flush all windowed register sets

ld [%i1+SP], %sp ! Load in ‘new thread’ handle

ld [%sp+L0_OFF], %l0 ! Start loading in ‘new thread’

ld [%sp+L1_OFF], %l1 ! ... register values.

...

Figure 8: Old register v alues and the new save area: a race condition

register set must b e marked as inv alid. The inv alidation is required so that when the new thread is started

and do es a pro cedure return, the register set of the new thread will b e loaded. If the register sets were left

v alid, then a pro cedure return in the new thread would access the register sets of the old thread.

Because the ST FLUSH WINDOWS trap saves register sets by using the stack p ointer, it must b e used b efore

the new thread’s stack p ointer is loaded.

7 Doing Double Dut y with the Sa v e Area

If cswap is implemented as a non-leaf pro cedure, the current register set will b e ushed out to memory by

the ST FLUSH WINDOWS trap, even though none of the current register v alues need to b e saved. If cswap is

a leaf pro cedure , the current register v alues need to b e ushed to memory and ST FLUSH WINDOWS will ush

the current register v alues, but will put them in the kernel window save area, which may b e overwritten

asynchronously .

This section describ es how to use ST FLUSH WINDOWS to write the register v alues only once, in the window

save area.

7.1 Keeping Sa v ed Register V alues Safe

It is tempting to try to use the kernel save area since ST FLUSH WINDOWS will write the current register set

to memory along with the other register sets. Doing so, however, is nontrivial. Consider the hyp othetical

context switch fragment in Figure 8 and the corresp onding v alue lo cations just after the new stack p ointer

has b een loaded into %sp, shown in Figure 9.

ST FLUSH WINDOWS saved the old register v alues to the kernel save area of the old stack, and we are

ready to pick up the new register v alues from the new stack. If, however, an interrupt arrives b efore the

new register v alues are read in from the new stack, the current register v alues will b e saved wherever the

stack p ointer p oints to. Thus, the old register v alues are going to overwrite the new register v alues. If the

interrupt came after %l0 had b een loaded, then the new v alue of %l0 would get written back, which is �ne,

but the rest of the new thread register v alues would get clobb ered.

There is, however, something we can do: we can save the old register v alues and then set the stack p ointer

%sp to a third kernel window save area that will only b e used if an interrupt happ ens when the new register

v alues are b eing reloaded. Then we load the new register v alues and set the stack p ointer to the correct ‘new

stack’ v alue. The imp ortant p oint is that while the registers are b eing reloaded, %sp cannot p oint at either

| | | |

| | | | CPU

+------+ +------+ +------+

| old | | new | | old |

| save | | save | | reg |

| vals | | vals | | vals |

+------+ +------+ <-sp +------+

Figure 9: Old register v alues and the new save area: a race condition

st %sp, [%i0+SP]

ta ST_FLUSH_WINDOWS

ld [%i1+SP], %o0 ! ‘%o0’ -- chosen arbitrarily

set _kwsa, %sp ! Don’t clobber new reg values

ld [%o0+L0_OFF], %l0

ld [%o0+L1_OFF], %l1

...

ld [%o0+I7_OFF], %i7

!

! All important values have been read from the ‘new’ kernel

! window save area, so now the kernel window save area can

! safely be overwritten if there’s an interrupt or whatever.

!

mov %o0, %sp

Figure 10: Reloading registers without a timing race

the old stack or the new stack. That is, the stack p ointer must p oint at a third area whenever the register

v alues are inconsistent with the saved v alues in the kernel window save areas of the old and new threads.

Figures 10 and 11 show race-free versions of Figures 8 and 9.

7.2 Selecting A T emp orary Register

In Figure 10, the choice of %o0 is mostly arbitrary

. It would seem that we could use %l7 as shown in

Figure 12. However, using one of the registers that is b eing saved or restore d from a save area (the in and

lo cal registers) creates a race condition.

Register %l7 is used as a temp orary and then gets its new v alue loaded when it is no longer needed as

a temp orary

. However, there is a slightly subtle race condition: if there is an interrupt after the old v alue

of %l7 is clobb ered and b efore %sp is set to the magic kernel window save area, then the saved v alue of %l7

in the old save area will get clobb ered as well. Therefore, in order to use %l7 it is necess ary to reverse the

order of the ld and the set, as shown in Figure 13.

Note that a similar situation exists when the v alue of %l7 is reloaded. It is tempting to set the stack

p ointer and then reload %l7. However if an interrupt o ccurs b etween setting the stack p ointer and loading

%l7 then the new thread v alue of %l7 will get clobb ered. Figure 14 shows an equiv alent race condition for

restoring registers .

When no temp orary registers are av ailable, the correct usage is to save the new thread stack p ointer to

memory and reload it again after restoring %l7. Memory must b e accesse d using the stack p ointer, b ecuase

no other registers are av ailable to form addresse s. The third kernel window save area must b e made slightly

| | | |

| | | | CPU

+------+ +------+ +------+ _kwsa: +------+

| old | | new | | old | |magic |

| save | | save | | reg | | save |

| vals | | vals | | vals | | area |

+------+ +------+ +------+ +------+ <-sp

Figure 11: Reloading registers without a timing race

ld [%i1+SP], %l7 ! Load an intermediate value

set _kwsa, %sp ! Now can clobber ‘in’ and ‘local’

! regs w/o clobbering saved copies.

ld [%l7+L0_OFF], %l0

ld [%l7+L1_OFF], %l1

...

ld [%l7+L7_OFF], %l7 ! Load new value

Figure 12: A new race condition: using %l7

set _kwsa, %sp ! Now can clobber ‘in’ and ‘local’

! regs w/o clobbering saved copies.

ld [%i1+SP], %l7 ! Load an intermediate value

Figure 13: Fixing the preced ing race condition

...

ld [%l7+L6_OFF], %l6

mov %l7, %sp ! Set ‘new thread’ stack pointer

ld [%l7+L7_OFF], %l7 ! Set ‘new thread’ %l7

Figure 14: Race condition when restoring registers st %l7, [%sp+64] ! Save ‘new thread’ %sp to ‘spsave’

ld [%l7+L7_OFF], %l7 ! Restore %l7

ld [%sp+64], %sp ! Restore ‘new thread’ %sp

Figure 15: Reloading new thread’s %sp from %sp

+------+

|spsave|

_kwsa: +------+

|magic |

| save |

| area |

+------+ <-sp

Figure 16: Reloading the new thread’s %sp from %sp

larger, as shown in Figures 15 and 16.

The follo wing section (x 7.3) develops a b etter way to restore registers in register-cramp ed situations.

Note that the out registers are usually free, but not if the cswap routine is inlined in the caller. Even if

the out registers are unav ailable, %g1 is nearly always av ailable

5

.

The prece ding example shows that cho osing the right register for holding the intermediate new stack

p ointer can b e tricky b ecause of p otential race conditions. F or many applications, some register will b e free

(e.g., %g1) avoiding the need to do double duty with one register.

7.3 Signals

The third kernel window save area lab eled kwsa needs to b e larger than 64 bytes b ecause it is b eing p ointed

to by the stack p ointer. On pro cess signals (which may happ en asynchronously), signal handlers may b e

pushed on the user stack ([CSR 86]; see b oth signal(3C) and sigstack(2)). Thus what we really want to do

is to allo cate the third kernel window save area on a stack. As shown in Figures 17 and 18, we can do that,

and put it either on the old stack or the new stack, just b eyond the kernel save area where the register v alues

got ushed by the ST FLUSH WINDOWS trap.

Figures 17 and 18 show the save area on the old stack. If registers are a problem as describ ed in Section 7.2

then the stack p ointer can b e used to restore the general register v alues. Then, the new thread v alue of the

stack p ointer is recovered by incrementing the stack p ointer instead of the rigmarole shown in Figure 15.

7.4 T rap Handler Con v en tions

The trap handler saves register v alues on the stack in a particular order. The user co de that restores register

v alues must use the same conventions as the trap handler. The o�sets in the kernel window save area

are de�ned (in at least some versions of SunOS) in <sun4/asm linkage.h>. Indeed, one might #define

KERNEL and try using RESTORE WINDOW(%o0), although co ding the restore by hand will b e faster if some of

the register restores can b e avoided.

5

The ma jor exception is for cases when a thread is b eing restarted from an asynchronous (preemptiv e) stop. In that case,

there are are also problems jumping back to the lo cation from which the program was stopped, since b oth the �nal register and

the jump address register must b e restored after the jump. The problems can b e solved using save and restore, alb eit at the

exp ense of using an extra register set.

st %sp, [%i0+SP] ! Save old sp.

ta ST_FLUSH_WINDOWS ! Save current regs.

!

! Don’t change ANY ‘in’ or ‘local’ registers until ‘sp’

! is updated, or the changed registers may get saved to

! the ‘old’ area, clobbering whatever was there.

!

sub %sp, 64, %sp ! Allocate a third kwsa

ld [%i1+SP], %o0

ld [%o0+L0_OFF], %l0

ld [%o0+L1_OFF], %l1

...

ld [%o0+I7_OFF], %i7

mov %o0, %sp ! Load new sp.

Figure 17: Context switch with kwsa on the stack

| | | |

| | | | CPU

+------+ +------+ +------+

| old | | new | | old |

| save | | save | | reg |

| vals | | vals | | vals |

+------+ +------+ <-o0 +------+

| |

| kwsa |

| |

+------+ <-sp

Figure 18: Context switch with kwsa on the stack

8 P erformance Considerations

The straightforward implementation of cswap allo cates a new register set (using save) and then manually

saves and restores the few v alues needed to restart the thread, namely %fp and %i7. The allo cation of an

extra register window will cause more over ows and under ows, and the ST FLUSH WINDOWS trap ushes the

freshly-allo cated register set to memory { even though it contains garbage v alues.

T urning cswap into a leaf pro cedure can eliminate some window over ows and under ows. However there

is still an extra window write, since the register window is written twice { once for the manual save to the

user save area and once for the ta ST FLUSH WINDOWS save to the kernel save area.

Using the kernel save area to hold register v alues (and allo cating a third kernel window save area) is

fastest b ecause the register v alues are written to memory only once, using the trap. As a b onus, the registers

do not need to b e saved manually

. In addition, if globals, caller-save, and oating-p oint registers don’t need

to b e saved, several stack adjusts can b e avoided.

Inlining the cswap routine eliminates pro cedure call and return overhead. However, replicating the context

switch co de slightly increases the total program size, and at each context switch the program counter must

b e saved explicitly since the context switch can happ en at any of several sites.

9 Conclusions

On the SP

ARC architecture, register windows act as a cache to hold register sets that are logically a part of

v arious stack frames. The register sets must b e ushed on thread context swap, but access to the register

window control registers is restricte d to kernel-privilege pro cess es . Thus, a threads pack age must make

careful use of the kernel primitives to save register windows.

This do cument shows in detail how a threads pack age can b e implemented for the SP

ARC if the machine

is running either SunOS or an op erating system that provides equiv alent functionality for ushing register

windows.

10 Ac kno wledgemen ts

Dirk Grunwald help ed with the user-space co de. Chandu Thekk ath help ed me to understand the kernel

co de.

References

[Anderson et al. 89] T. E. Anderson, E. D. Lazowsk a, and H. M. Levy

. The p erformance implications of

thread management alternatives for shared memory multipro cessors. IEEE T r ansactions on Com-

puters, 38(12), Decemb er 1989.

[Bershad et al. 88] B. N. Bershad, E. D. Lazowsk a, and H. M. Levy

. PRESTO: A system for ob ject-oriented

parallel programming. Softwar e { Pr actic e and Exp erienc e, 18(8), August 1988.

[Birrell 89] A. D. Birrell. An intro duction to programming with threads. T echnical rep ort, Digital Equipment

Corp oration Systems Research Center, 130 Lytton Avenue, Palo Alto, California 94301, 1989.

[CSR 86] CSRG Computer Science Division, Department of EE/CS, University of California, Berkeley ,

California 94720. Unix Pr o gr ammer’s R efer enc e Manual, 1986.

[Fischer & LeBlanc 88] C. N. Fischer and R. J. LeBlanc, Jr. Cr afting A Compiler. Benjamin/Cummings,

1988.

[Maturana 91] G. Maturana. Personal communication, April 1991.

[Sun 91] Sun Microsystems. The SP

ARC Ar chite ctur e Manual, V ersion 8, 1991.

[Thacker et al. 88] C. Thacker, L. Stewart, and E. Satterthwaite Jr. Fire y: A multipro cessor workstation.

IEEE T r ansactions on Computers, 37(8):909{920, August 1988.

[Y oung et al. 87] M. Y oung, A. T ev anian, R. Rashid, D. Golub, J. Eppinger, J. Chew, W. Bolosky

, D. Black,

and R. Baron. The duality of memory and communication in the implementation of a multipro-

cessor op erating system. In Pr o c e e dings of the 11th ACM Symp osium on Op er ating Systems

Principles, pages 63{76, Novemb er 1987.

0 1 2 3

+---+---+---+---+

| | 1 | | | WIM

+---+---+---+---+

+---+---+---+---+

| | | U | |

+---+---+---+---+

^CWIN

Figure 19: Window Inv alid Mask (WIM) and register sets

A Register Windo ws in the Kernel

This section discusse s kernel concerns for register sets and is not directly related to user-space threads.

A.1 Basic Op eration

There are two register vectors in the CPU. One is a vector of register sets, where each register set consists

of the in and lo cal registers. The other vector is a bit vector called the WIM (window inv alid mask) saying

which registers are inv alid, where \inv alid" means cannot b e used by a user pro cess without �rst saving some

other register set.

The save instruction p erforms like an sub instruction except that it tries to decrement the CWIN (current

window) p ointer. If the bit is set for the new register (that is, WIM[CWIN] == 1), then the decrement fails,

and the pro cess or traps.

In Figure 19, U is the user register set <<1>>, and a save instruction will cause a trap. The trap will

automatically decrement the CWIN by 1. Even though CWIN now p oints to an \inv alid" register set, a second

trap will not result. The kernel now has an \empty" register set <<1>> that it can use while it is saving

other registers. In particular, some of the global registers can b e moved into lo cal registers in <<1>>, so

that the kernel can use the global registers.

When the kernel executes a save instruction, that decrements CWIN to zero. It must b e that register

window 0 needs to b e saved, or else the WIM for <<1>> would not have b een set. In the <<0>> register

window, one of the registers is the stack p ointer. The kernel checks that the stack p ointer p oints to a v alid

place to save the registers. If not, it causes the user pro cess to take an exception (such as a segmentation

fault, bus error, etc.). If the stack p ointer do es p oint to a v alid place to save registers, then they are simply

written. Indeed, the kernel can continue to march through the register windows, saving <<3>> and <<2>>

as well. At some p oint, the kernel will have cleared \enough" register sets and will use restore to return to

the prop er register set (<<1>>), restore the v ariables that it stored away at the b eginning, and then return

control to the trapping pro cess with a jmpl and a rett.

A.2 Boundary Conditions

Each register set contains a frame p ointer which p oints to the save area for the next register set. The register

set do es not contain the address of its own save area. That is, in the current window %o6 is the save address

for the register set i0..i7 and l0..l7.

The normal implementation of the window overflow trap handler is to save some register sets and then

restart the save instruction that caused the trap. A SP

ARC implementation with only two register sets

[Sun 91], is shown in Figure 20. The trap handler can set CWIN to zero or one, and can set WIM to either 01

or 10. Every p ossible combination, however, will cause problems. If the CWIN and the v alid register set are

0 1

+---+---+

| 1 | | WIM

+---+---+

+---+---+

| | U |

+---+---+

^CWIN

Figure 20: Two register sets

the same, then restarting the save instruction will cause the save instruction to trap. If CWIN and the v alid

register are di�erent, then the rett in the trap handler will cause a window underflow trap. Thus, it is

necess ary to emulate the save instruction in software and restart the user co de at the instruction follo wing

the save [Maturana 91].

B Cac hing Threads in Register Windo ws

In the preceding sections, the register windows are allo cated entirely to the currently-exec uting thread. On

a context swap, all active register sets are ushed. Thus, a context swap takes at least as long as the time

to save the sixteen registers in one set and may b e as long as the time to save sixteen registers p er set times

the numb er of sets. Thus, the context switch time may b e b oth long and v ariable.

An alternative is to allo cate just one window to a thread. Pro cedure calls and returns must either manage

registers explicitly (as with a traditional, non-windowed register set) or must incur a trap on each save and

restore. Supp ose there are four av ailable register sets and that each register set holds the registers for an

active thread. Then, context switches b etween any of those threads can b e p erformed with a small numb er

of save or restore instructions. If an inactive thread needs to b e run, one of the active register sets is

ushed to memory and the registers for the inactive thread are loaded. There is still large v ariability but (a)

the absolute time is never more than the time to save and restore 32 registers; and (b) an imp ortant thread

can b e \hard" allo cated to a particular register set, so that it is always av ailable in short order.

B.1 Implemen tation Considerations

Although the scheme descr ib ed ab ove is simple, there are several complications. First, global registers are

not windowed, so they may need to b e saved and restored explicitly

. Second, each register set holds only

sixteen registers (in and lo cal registers); furthermore, each active register set must have an adjacent register

set av ailable for trap handlers. Thus, each thread takes two register sets. Third, the movement b etween

register sets (using save and restore) must b e p erformed in the kernel. If save and restore were executed

in user space, the current window p ointer would sometimes p oint at one of the register sets reserved for

traps, faults, and interrupts. If an exception o ccured when the current window was one that was \saved

for exceptions", then the exception handler would use (clobb er) an adjacent window that is used for thread

registers. F ourth, an implementation may have an o dd numb er of register sets, so that register sets cannot

b e paired for user/exce ption use in a straightforward way .

A user-space threads pack age should then have av ailable at least: (a) enough information to determine

how many register set pairs are av ailable; and (b) one or more kernel entries that adjust the current register

set. The user-space pack age is resp onsible for determining whether or not a thread is cached and for saving

and restoring as necess ary if the needed thread is not currently cached.

_cswap:

std %i0, [%sp+64] ! ! 2

std %i2, [%sp+72] ! ! 2

std %i4, [%sp+80] ! ! 2

std %i6, [%sp+88] ! ! 2

std %l0, [%sp+96] ! ! 2

std %l4, [%sp+104] ! ! 2

std %l6, [%sp+112] ! ! 2

std %l8, [%sp+120] ! ! 2

st %sp, [%o0] ! Save old handle ! 1

ld [%o1], %o1 ! Get new handle ! 1

sub %o1, 64, %sp ! Go to new thread ! 1

! ... allocate kwsa !

ldd %i0, [%sp+64] ! ! 2

ldd %i2, [%sp+72] ! ! 2

ldd %i4, [%sp+80] ! ! 2

ldd %i6, [%sp+88] ! ! 2

ldd %l0, [%sp+96] ! ! 2

ldd %l4, [%sp+104] ! ! 2

ldd %l6, [%sp+112] ! ! 2

ldd %l8, [%sp+120] ! ! 2

retl ! Return from cswap ! 1

add %sp, 64, %sp ! Deallocate kwsa ! 1

! Total: 37

Figure 21: Manually spill and reload registers

F or example, an implementation might allo cate an extra curr ent window word in the thread’s state, where

the current window word is negative if the thread is inactive, or holds the current register set pair if the

thread is in a register set. On a context swap, the context swap routine checks the current window word. If

negative, the current register set is dump ed and the new thread is loaded into the current register set. No

kernel traps are needed. If the current window word is non-negative, then a go window kernel entry is called

with a single argument naming the register pair to b e used next.

In the follo wing subsect ions, we provide (untested) sample implementations so that we can analyze the

cost of allo cating one window p er thread. These sample implementations will give a metho dology for deciding

what implementations are pro�table, whether the threads pack age should simply save and restore registers

or ignore register windows altogether.

In the following analysis, we assume that the global and oating-p oint registers are caller-save. W e

further assume that loads and stores take a single cycle or two cycles for a double-precision load or store

(though the result of a load is not av ailable in the following cycle) and that branches take a single cycle

(though there is a delay slot).

B.2 Alw a ys Spill and Reload

The co de in Figure 21 shows what is needed to simply save and restore the registers manually

. The costs

for loading and storing v alues assumes that the v alues are in-cache and assumes a fast cache. Real times are

probably somewhat worse. Our estimate is 37 cycles.

B.3 Cac hing Threads

_cswap:

ld [%o1], %o1 ! Get new thread handle ! 1

ld [%o1+64], %o2 ! Find new thread’s window # ! 1

sethi %hi(_win), %o5 ! Find current thread’s ! 1

ld [%lo(_win)+%o5], %o3! ... window number ! 1

sub.cc %o2, 64, %g0 ! If new thread in window ! 1

jg new_window ! ... do window case ! 1

! Delay slot, fill w/ ‘ld’ ! 0

! Total: 6

/* Not in a window. */

! ‘_win’ not changed ! 0

sub %sp, 72, %sp ! Alloc. space to save win # ! 1

! ... and registers ! 1

... ! Save 16 regs ! 16

sub %g0, 1, %o3 ! Create invalid window # ! 1

st %o3, [%sp+64] ! Save invalid window number ! 1

st %sp, [%o0] ! Remember old thread ! 1

sub %o1, 64, %sp ! Alloc kwsa, switch ! 1

... ! Restore 16 regs ! 16

ret ! Restart new thread ! 1

add %sp, 72, %sp ! Dealloc regs save area, ! 1

! ... and window number

! Total: 40

/* Go to the new thread’s window. */

new_window:

st %o3, [%lo(_win)+%o5]! Save new window number ! 1

sub %sp, 8, %sp ! Alloc space to save win # ! 1

st %o3, [%sp+64] ! Save valid window number ! 1

st %sp, [%o0] ! Remember old thread ! 1

ta %go+go_window ! Change windows (all regs!) ! 3

retl ! return from cswap ! 1

add %sp, 8, %sp ! Deallocate window number ! 1

! Total: 9

Figure 22: Context switch by switching register sets

/* Did one ‘save’ already when we trapped. */

TRAP_go_window:

and %o2, 0x1e, %o2 ! Limit length of loop and ! 1

! ... prevent goto odd win # !

rd %psr, %o4 ! Get current %psr register ! 1

and %o4, 0x1f, %o4 ! Extract curr. win # ! 1

sub %o2, %o4, %o4 ! Form distance to new win # ! 1

sub.cc %o4, 2, %o4 ! Two closer than we were ! 1

jnz L0 ! Go again if more windows ! 1

save ! Delay slot: second ‘save’ ! 1

jmpl %l2 ! Return from trap ! 1

rett ! Return from trap ! 1

L1: save ! Move over by one window ! 1

L0: save ! Move over by one window ! 1

sub.cc %o4, 2, %o4 ! Two closer than we were ! 1

jnz L1 ! Go again if more windows ! 1

! Fill w/ L0’s ‘save’ ! 0

save ! Instead of restore! ! 1

jmpl %l2 ! Return from trap ! 1

rett ! Return from trap ! 1

Figure 23: T rap handler to switch register sets

The co de in Figure 22 is the user-side co de for a context swap. The user-side co de makes use of a

‘go window’ trap handler shown in Figure 23. The trap handler assumes that there are an even numb er

of registers. Some SP

ARC implementations have an o dd numb er, in which case the trap handler is more

complicated and takes longer to execute.

The trap handler shown here only moves through the register set in one direction. On an 8-window

system, if the current user window is 3 and the desired window is 5, the trap handler will cycle through

windows 1, 0, 7, and 6 to get to 5. This is ab out the same cost as cycling through 3 and 4 to get to 5. If,

however, the system has 16 windows, the move from 3 to 5 will require cycling through 1, 0, 15, 14, ..., 7,

and 6, which is much more exp ensive. The handler can b e optimized by deciding whether to use save or

restore to go to the correct register set. However, the initial decision may b e exp ensive.

Note also that threads may thrash. When a thread \misses" in the windows cache, it is loaded in to the

same window as the most-recently executing thread, thus the old thread must b e forced out to memory

. If

the threads alternate execution, then every context switch will miss. An alternative is to increment to the

next register set, but at an additional cost of 9 cycles, raising the miss cost from 46 cycles to 55 cycles.

B.4 Comparison

If threads are not cached in windows, the average path is 37 cycles. If threads are cached, but the thread

is not in a window, the average path is 46 cycles. When a thread is in a window, then the path through

cswap is ab out 15 cycles. In addition, the kernel co de for go window is executed. The cost to move over

one register pair is 9 cycles. T o move over two pairs takes 12 cycles, three pairs 16 cycles. Three pairs is

the maximum p ossible distance in any current SP

ARC implementation. Assuming all move distances are

equally probable, the average move distance is (24 + 27 + 31)=3 = 27:3 cycles.

Thus, if threads \hit" in the windows an average of P

hit of the time, then the two metho ds (cache threads in windows and don’t cache threads in windows) take the same time when Equation 1 is satis�ed. F or the

estimated times ab ove, the breakeven is when 27:3 � P

hit

+ 46 � (1 � P

hit

) = 37 or when ab out 50% of all

thread context swaps are to threads that are cached.

hit cost � P

hit

+ miss cost � (1 � P

hit

) = noncached cost (1)

Moving over by 16 register set pairs (which is the maximum p ossible in any p ossible SP

ARC implemen-

tation) would take 4 � 15 + 9 = 69 cycles in the handler, a total of ab out 84 cycles. Even if the average

distance to move was only 8 register set pairs, the average cost would b e 52 cycles, more than the cost of

staying in one window.

Note that these numb ers are derived using estimated times. The real times are implementation-dep endent.

F or instance, memory accesse s may b e (relatively) more exp ensive on an implementation with a high clo ck

sp eed, while branches may b e more exp ensive on a machine with lo ok ahead. Thus, it may b e imp ortant that

the user-space-only co de is memory intensive but executes straight-line co de and that the cached windows

co de is register-to-registe r intensive but has lots of branches.